
Overview

XFRX is a powerful third party tool for allowing you to send your FRX report files to various output formats.  The full version allows you to create reports in PDF, Word, HTML, Excel, RTF, XML, various kinds of image files, and other formats. Output to PDF, HTML and the XFRX internal XFF formats also allow the creation of bookmarks.  This session will focus on what it takes to get XFRX working with an existing application, and will specifically focus on the PDF, Word and HTML output types.  We’ll look at how the use of a Report Class helps with using third party products like XFRX, and allows you to switch tools simply by changing a property.
What is XFRX?
XFRX is a third party product that can take reports you’ve created as FRX files in VFP and render them in any of a number of output formats.  Although you pay an annual maintenance fee to receive upgrades, XFRX includes a royalty-free runtime license.  The license restrictions only state that you may not distribute the source code, and you may not incorporate it into a framework or report generation product that you sell to application developers.
You may download the evaluation version and can review the license and manual online at http://www.eqeus.com.  You may also buy the product online at this site. Purchase options include either a single output format only at a reduced price or all output formats at a very affordable price.  (For current pricing, please visit the web site.)  A source code option is also available.

Without the source code option, you get an XFRX.FXP instead of source code, but you still get source code for the class library, which includes the Preview Container and a form with the Preview Container.

XFRX will run and produce all output formats in VFP versions all the way back to 5.0.

Martin Haluza is the creator of XFRX and he provides top-notch support for the product.  I’ve found him very responsive to both bug reports and enhancement requests.

Caveats

If you intend to create Word output, you must have Word 2000 or later on your machine.
I’ve had some problems using Word output with XFRX on a machine that is running McAfee Antivirus.  McAfee seems to think that VFP is running a malicious script, and even after you tell McAfee to let the script run, it appears to hang VFP after the report is complete because of a dialog that you cannot see.  Martin says the problem is that he uses COM to create Word and Excel files and McAfee is configured by default not to allow COM.  A change in the McAfee configuration will fix this problem.  Depending on your hardware configuration, this could mean adding your IP address to a list of trusted addresses or adding your LAN to a list of trusted connections.  Output formats other than Word and Excel are done natively and don’t have this problem.
A little history

My Report Class, which is essential to this presentation, came about as a result of a need for a client.  This project was in VFP 6, the latest version at that time.  The project had a central screen from which I generated reports.  The user could select up to fifteen reports and they would all be generated.  The one big problem with this was that if the user wanted to preview the reports, they would have to close the preview for one report before being able to view the next.  And of course, they couldn’t go back to the previous report.

The client requested the ability to generate the reports in PDF format.  We needed to find a tool that didn’t require massive changes to the existing reports and one that allowed us to add multiple reports to the same PDF file.  I did a bit of research and selected the Mind’s Eye Reporting Engine.  The product was still in beta at the time and not all my reports worked with it.  I found myself having to switch code back and forth for individual reports to run each report in native VFP versus the Mind’s Eye Report Engine.

It became clear that I needed to move the mechanism for generating reports into a common and generic set of code.  That’s when I created the Report Class.  I included a property that could be changed at runtime to indicate if a report should be run using native VFP or the Mind’s Eye Report Engine.  This allowed me to run the reports that worked with Mind’s Eye in the Mind’s Eye Report Engine, and those that didn’t could be run using native VFP.  Of course, the ones that couldn’t run under Mind’s Eye wouldn’t be included in the PDF and would preview or print before the PDF.
Soon my client needed to get a release of their product out, but Mind’s Eye still had some problems that hadn’t been worked out yet.  I needed a different tool that didn’t have the problems we had with Mind’s Eye.  After asking around some more, I was told about XFRX.  XFRX had the advantage of being able to generate more output formats than Mind’s Eye, but the important thing was that it didn’t have the problems with some of my reports that Mind’s Eye did. 

Having the Report Class in place made switching over to using XFRX very easy.

Before we begin

There are some things you’ll need installed on your computer before you can run the sample code.  You must be running VFP 9 to run the code in my sample.  Although XFRX will work with previous version of VFP, the techniques used in VFP 9 are different from previous versions.  The XFRX manual documents techniques for both VFP 9 and previous versions, and my Report Class can easily be modified to use the older technique.  You must also have version 10.2 or later of XFRX in order to print from the preview.  This is because the PrintDocument method wasn’t introduced until version 10.2.  I strongly recommend that you use the latest version of XFRX however, because there were a number of bugs with the printing from the preview in version 10.2 that were fixed in later versions.
The Report Class has the code necessary to run reports in native mode, XFRX or the Mind’s Eye Report Engine, although most of the code for the Mind’s Eye Report Engine is commented out.  I’ve included the evaluation version of XFRX with the source code for the session, but not an evaluation version of Mind’s Eye.  If you would like to try the Mind’s Eye evaluation, you can download it from http://www.mindseyeinc.com/  I have not tested this code with Mind’s Eye in a while, so the latest version of Mind’s Eye may not be backwards compatible with the version I had when I wrote that part of the Report Class.
IMPORTANT!  When you unzip the code, make sure you retain the relative paths of the files or the program won’t work.

I have not added comments in my code for XFRX method calls that are well documented in the XFRX manual.  Properties that are not well documented in XFRX I’ve tried to document here or in my code.

Getting XFRX to work during development

The default path for XFRX when you unzip it is C:\XFRX.  For the purpose of this demo, I’ve put the XFRX directory directly under the main project directory instead.  Either way, you need to SET PATH TO the main XFRX directory and the XFRXLIB directory under it. 
XFRX also comes with a file called XFRX.FXP, which you must include in your project.  If you are running the evaluation version, you’ll have an XFRX.APP file instead of the FXP file.  
This is all you need to do to make XFRX available to your app.  You can now instantiate XFRX and call its methods. 

The Demo program

The demo program included with this session demonstrates how to integrate XFRX into an app using a Report Class.  The project is called “Demo” and the main program is Main.prg.  Running Main.prg will show you the form shown in Figure 1.

Figure 1 – The main screen of the Demo program
The Demo program allows you to generate reports using either native VFP or XFRX.  The PDF, Word and HTML output formats are only available if you choose XFRX as the output tool.
I’ve included three reports with the Demo program.  The “Score Sheet” is a single page, portrait report.  “Work in Progress Detail” is a multiple page, landscape report.  “Work in Progress Summary” is a single page, landscape report.  You may choose any one, two, or all three of these reports for your output, but they will always print in the order shown.
Using XFRX directly

I mentioned that I use a Report Class for handling my reporting.  I’ll demonstrate how I my Report Class uses XFRX directly.

Initialization

First, you must initialize XFRX.  You need to create a variable for holding a reference for XFRX that will stay in scope as long as you need it.  I have a property in my Report Class for this called oRpt.  The example below shows how I instantiate XFRX.

* For the evaluation mode, you need to wrap calls to the XFRX() method with 

* an EVALUATE() function or it won't compile.

* For the regular version, this isn't necessary.

this.oRpt = EVALUATE([xfrx("XFRX#LISTENER")])

The next thing XFRX requires is a call to the SetParams method, but the first parameter of this method is the output file.  Therefore, you want to figure out what the output file is before you call SetParams.

DO CASE 

CASE this.cOutput = 'PREVIEW'

   lcOutputFile = ''

CASE this.lDontConfirmOutput

   lcOutputFile = This.cDefaultOutputFile

OTHERWISE  

   lcOutputFile = PUTFILE(This.cOutput+ ;

       " file name:",This.cDefaultOutputFile,This.cOutput)

ENDCASE

IF NOT EMPTY(lcOutputFile) OR this.cOutput = "PREVIEW"

   lnRetVal = this.oRpt.SetParams(lcOutputFile, ;

              ,.F., ,.T., ,IIF(This.cOutput="PREVIEW",'XFF',This.cOutput))

   IF lnRetVal = 0


* Everything was Ok

   ELSE


MESSAGEBOX(This.oRpt.ErrorMessage(lnRetVal),16,"Report error")

   ENDIF 

ELSE

   This.lCancelProcessing = .T.

ENDIF

The parameters for SetParams are well documented in the XFRX manual, so I will not repeat that information here.  Note that in VFP you have the option of setting properties for the output file and the output file type, and may not need to send any parameters to the SetParams method.
Running Reports

The next step is to actually call your reports.  Below is code for a single report.
REPORT FORM SomeReport OBJECT This.oRpt noconsole

Note this is VFP 9 syntax, and we are sending a reference to the XFRX object, which is a report listener, to the report engine.  You must call this REPORT FORM command for each report you want to include in your output.  It is the REPORT FORM command that renders the reports.  
When running multiple reports, you probably want continuous numbering across the reports.  In order to do that, you need to include the NORESET clause with your REPORT FORM command.  If you have additional reports, you also want to include NOPAGEEJECT for all but the last report.

Between multiple REPORT FORM commands and the various report clauses you may need to use, you can see that a function that does this for you can be quite useful.  The aRpts2Run property of my Report Class stores the reports to be run and the XFRXOutput method runs the multiple REPORT FORM commands with the necessary clauses.  I cover this in more detail below.
The Report Class
The code for the Report Class is included in the source code provided with this session.  It is in Main.prg and is the class called DoReports.  For the simplest example of its use, you would instantiate the Report Class, tell it which reports to run, and output the reports.  The code to do this looks like this:

loRpt = CREATEOBJECT("DoReports")

* The "OutputClause" is where the report should go, i.e. "Preview", "Printer"

loRpt.InitRptMode(OutputClause)

Select SomeTable

loRpt.DoIt("ReportName1")

Select AnotherTable

loRpt.DoIt("ReportName2")

loRpt.SendRptToOutput()

That’s not too hard, is it?  Let’s take a closer look at what each line of code is doing for us.
After instantiating my report class, I call the InitRptMode method.  This will run the initialization code needed for XFRX or whatever output tool I decided to use.

I select a table or cursor, which must be the alias that drives the report.  Then I call the DoIt method, passing the name of the report.  For XFRX, this queues the report in the aRpts2Run array, storing the name of the report and the current alias.

After queuing up all the reports, I call the SendRptToOutput method.  SendRptToOutput will call my XFRXOutput method, which performs the REPORT FORM commands.  This code is shown below.
FOR lnRpt = 1 TO ALEN(This.aRpts2Run,1)

   SELECT (This.aRpts2Run[lnRpt,2])

   * Note: NORESET affects the page number that the CURRENT report starts with

   DO CASE 

   CASE lnRpt = 1

      * We want to make sure the page number is reset for the first

      * report, but only keep it from ejecting if there are more

      * reports to run.


IF ALEN(This.aRpts2Run,1) = 1


   lcClauses = ""


ELSE


   lcClauses = "NOPAGEEJECT"


endif

   CASE lnRpt = ALEN(This.aRpts2Run,1)

      lcClauses = "NORESET"

   OTHERWISE 


lcClauses = "NOPAGEEJECT NORESET"

   ENDCASE



   REPORT FORM (This.aRpts2Run[lnRpt,1]) OBJECT This.oRpt ; 

              &lcClauses noconsole

ENDFOR 


Report Class properties

Now it’s time to delve into this object a little deeper.  First, let’s take a look at the properties.  These are initialized at the top of the class definition.

oRpt = .Null.

lFirst = .T.

cSetClass = SET("Classlib")

cReportMode = "NATIVE"

cDefaultOutputFile = ''

lDontConfirmOutput = .F.

cPrinter = ''

cOldPrinter = ''

lCancelProcessing = .F.

lXFRXBookmarks = .F.

cProgressMess = ''

cOutput = ''

oScripts = .Null.

DIMENSION aRpts2Run[1]

The code above shows how each of these properties is initialized.  Here’s how they are used:

oRpt – Both XFRX and Mind’s Eye require that you instantiate an object in order to use them.  This property maintains a reference to the object needed for the product you’re using.  For native reports, this property stays NULL.

lFirst – This property is used internally by the Report Class and is only required by Mind’s Eye because the first time you call it with a report name it creates a form object, and subsequent calls require an Append type of command.  This property is ignored when using native or XFRX mode.

cSetClass – This property is used internally by the Report Class. Because we need to alter our SET CLASSLIB command for Mind’s Eye, I store the original setting in this property so we can set it back when the Report Class is destroyed.  This property is not used for XFRX or native mode.
cReportMode – This is the property you change when you want to change the tool being used for the current report.  Valid options at this time are “NATIVE” for native FoxPro reports, “XFRX” for reports rendered in XFRX and “MINDSEYE” for the Mind’s Eye Report Engine.

cDefaultOutputFile – You can define a default path and/or file name to use when creating output that generates a file, i.e. PDF, Word or HTML.

lDontConfirmOutput – If .T., the output file name will be whatever you’ve set the cDefaultOutputFile property to be, without asking the user first.  If lDontConfirmOutput is .F., the cDefaultOutputFile will be the default, but the program will use PUTFILE() to confirm the output with the user.  Note that by using PUTFILE() you cannot suppress the “File exists” message if the user selects a file that already exists.  PUTFILE() does not respect the setting of SET SAFETY.

cPrinter – This property is used internally by the Report Class and is used in native mode only.  It asks for the printer when you initialize the object for Print mode, and saves the selection for subsequent reports so that it doesn’t have to ask the user each time.  You may also change this setting between running reports in native mode if you want to redirect some reports to a different printer.

cOldPrinter – For native mode only.  The original printer selection is saved here so that it can be reset when the object is destroyed.

lCancelProcessing –This property is set to .T. when the user clicks on the Cancel button during confirmation of the output file.

lXFRXBookmarks – Determines if bookmarks are turned on or off when outputting from XFRX.

cProgressMess – You may specify a message to be displayed in a WAIT WINDOW with each call to the DoIt method or the SendRptToOutput method.

cOutput – This is what type of output to create, e.g. PREVIEW, PRINTER, PDF, etc.  When you call the InitRptMode method, you send this as a parameter and the value is stored in this property for use in methods called later.

oScripts – This is an object reference to the class defined at the end of Main.prg, called MyXFRXScripts.  This class is a place where you can define page bound scripts for XFRX.  I discuss this more later.
aRpts2Run (array) – This is a two-dimensional array that stores the set of reports to be run and the main alias for each report.  This is necessary for the preview and for all VFP 9 output.  Pre-VFP 9 uses the ProcessReport method, which doesn’t need to queue reports into an array.  See the comments in the demo code.
One thing you should be aware of with pre-VFP 9 versions of XFRX is that, it performs a queuing of the reports to be printed, but doesn’t actually render them until you give a command to finalize the report output.  The DoIt method of the Report Class shows this older technique in code that is commented out.  This older code performs the queuing by calling ProcessReport.  The new technique I use is to add the report to an array.  The DoIt method shows both techniques. For the older technique the SendRptToOutput method would be changed so it only calls the Finalize method.  If you are using the Mind’s Eye Report Engine, it figures out the rendering as each report is queued, and sends it to the selected output when you finalize.  The syntax is different for each of these tools, and both are different from native VFP syntax.  The Report Class hides this complexity from you by allowing you to use the same syntax, regardless of which tool you are using for the output.  You simply change the cReportMode property sometime after initializing the object.

The Report Class in use
Let’s look at an example of using the Report Class.  First, look at Main.prg.  This is a simple example of a main program used to start an app.  The important part of the main section of this program is where it sets the path setting we need to run XFRX.  It then calls the PrtOut form.
* Must use absolute pathing for cases when they send output to a PDF in a 
* different directory.

* Acrobat is changing the default directory!

oApp.AppDir = SYS(5)+CURDIR()

set path to XFRX;XFRX\XFRXLIB

IF NOT DIRECTORY('REPORTS')

   MD REPORTS

ENDIF 

do form PrtOut

Because of the way I’ve designed this demo, all reports should use a single output tool.  It is not difficult, however, to change the cReportMode property while processing multiple reports.  If you decide to do this, you also must change the initialization process so that it doesn’t rely on a single InitRptMode call.  Keep in mind that any reports you print in native mode will print or preview before all other reports.
PrtOut form’s Init method

Open the PrtOut form from the sample code and look at the Init method.  You’ll see here that I’m using a form property called oRptCls to store an object reference to the Report Class.  Because the DoReports code is in Main.prg, and Main.prg is what calls this form, I don’t need a SET PROCEDURE TO MAIN in order for the form to find the DoReports class.
In the Main.prg, I created a subdirectory called REPORTS if it didn’t already exist.  In the Init of the PrtOut form, you can see that I’m setting that as the default output location.  The default output location can be a directory or a specific file.  For this demo, I’m allowing the default file and location to be changed by the user at runtime.  
USE Scores IN 0

USE WIP IN 0

USE WIPTotals IN 0

thisform.oRptCls = CREATEOBJECT("DoReports")

Thisform.oRptCls.cDefaultOutputFile = "REPORTS\"

The code in the Report Class uses the PUTFILE function to determine the output file name and path.  I use PUTFILE instead of GETFILE because PUTFILE allows us to specify a default directory and file name.  However, PUTFILE doesn’t respect the SET SAFETY setting, so even if you have SET SAFETY OFF and choose an existing file, PUTFILE will display a message that the files exists and will confirm that you want to overwrite it.
PrtOut form’s OK button’s click method

In the OK button’s click method, I initialize a variable called lcRptClause.  The lcRptClause variable is based on the output format, i.e. “PREVIEW” for Screen, “PDF”, “DOC” for Word, etc.

The Report Class’s InitRptMode method is then called, passing it the lcRptClause.  If you are using native mode and are sending reports to the printer, the app will prompt you for the printer at this point.  If you are using XFRX for your output and are sending reports to a file, the app will prompt you for the file at this point, unless you have set lDontConfirmOutput to .T.  If you hit the Cancel button in either mode from the output prompt, the Report Class’s lCancelProcessing property is set to .T.  The code in the click method checks this property to see if processing of reports should continue.
LOCAL lcRptClause

IF thisform.oRptCls.cReportMode = "MINDSEYE"

   MESSAGEBOX("This demo isn't setup to run Mind's Eye")

   return

ENDIF 

DO CASE 

CASE thisform.opgOutput.value = 1

   lcRptClause = 'PREVIEW'

CASE thisform.opgOutput.Value = 2

   lcRptClause = "PDF"

CASE thisform.opgOutput.Value = 3

   lcRptClause = "DOC"

CASE thisform.opgOutput.Value = 4

   lcRptClause = "HTML"

CASE thisform.opgOutput.Value = 5

   lcRptClause = "PRINT"

OTHERWISE

   MESSAGEBOX("You must specify where to send reports!",48,"Missing option")

   RETURN 

ENDCASE

thisform.oRptCls.InitRptMode(lcRptClause)

IF NOT Thisform.oRptCls.lCancelProcessing

   thisform.PrtReports()

ENDIF 

Note that in the Mind’s Eye Report Engine you don’t need to know the type of output nor the name of the output file until you finalize the reports.  When outputting to the Mind’s Eye Report Engine, the Report Class won’t ask for the output file until you call the SendRptToOutput method.  With XFRX, this must be done as part of the initialization process.  The Report Class saves you from having to be concerned about these details and calls PUTFILE at the appropriate time for the output tool you’re using.
PrtOut form’s PrtReports method

The PrtReports method is a custom method that shows what it takes to run the three reports available in this demo program.  The essential part of this method checks the value of the checkbox for each report to see if you’ve selected the report.  If so, it sets the cProgressMess property, which will display a message for you in a WAIT WINDOW while the DoIt method is processing.  If cProgressMess is an empty string, which is the default, no message will be displayed.  You may find that in many cases, your reports will process quickly enough that you don’t want the display.
Next, I call the DoIt method, passing the name of the report.  The file that drives the report must be selected at this point!  If you are using native mode, the report will print or preview immediately.  For other modes, this queues the report.
Finally, I call the SendRptToOutput method.  If you are using native mode, this method doesn’t do anything.  For other report modes, the reports are sent to the output at this point.  For this reason, you must make sure that you haven’t closed any of the cursors needed for the reports before this point.  Below is the code in the PrtOut method of the form.
LOCAL lnChkCnt

lnChkCnt = 0

FOR EACH loObject IN Thisform.Controls

   IF UPPER(loObject.Baseclass) = "CHECKBOX"

      IF loObject.Value = 1

         lnChkCnt = lnChkCnt + 1 

      ENDIF 

   ENDIF 

ENDFOR 

DO CASE 

CASE lnChkCnt = 0

   MESSAGEBOX("No reports selected",64,"No selections")

   RETURN 

CASE lnChkCnt = 1

   Thisform.oRptCls.lXFRXBookmarks = .F.

OTHERWISE

   Thisform.oRptCls.lXFRXBookmarks = .T.

ENDCASE 

IF Thisform.opgOutput.value <> 0

   IF thisform.chkScores.Value = 0 AND Thisform.chkWIPDetail.Value = 0 ;

   AND thisform.chkWIPSumm.Value = 0

      MESSAGEBOX("No reports selected")

   ELSE 

      IF TYPE("thisform.oRptCls") = "O"

         IF thisform.chkScores.Value = 1

            * If you need to run code to build a cursor, call it here.

            Thisform.oRptCls.cProgressMess = "Generating Score Sheet"



SELECT Scores



thisform.oRptCls.DoIt("Scores")

         ENDIF 

         IF thisform.chkWipDetail.Value = 1

            * If you need to run code to build a cursor, call it here.



Thisform.oRptCls.cProgressMess = "Generating WIP Report"



SELECT WIP



thisform.oRptCls.DoIt("WIP")

         ENDIF 


   IF thisform.chkWIPSumm.Value = 1



* If you need to run code to build a cursor, call it here.



Thisform.oRptCls.cProgressMess = "Generating WIP Summary"



SELECT WIPTotals



thisform.oRptCls.DoIt("WIPTotals")

         ENDIF 

         Thisform.oRptCls.cProgressMess = "Finalizing reports..."


   thisform.oRptCls.SendRptToOutput()


   WAIT CLEAR 


ENDIF 
&& type("oRptCls") = "O"

   ENDIF 

ENDIF
&&  Thisform.opgOutput.value <> 0

* Set the default directory back to the app's directory in case it was 

* changed by the Acrobat Reader

CD (oApp.AppDir)

RETURN
If you are using XFRX or the Mind’s Eye Report Engine, any of the output methods will combine multiple reports into a single output file or preview.  Both portrait and landscape reports may be combined in this output.

Scripts in XFRX

Version 11 of XFRX supports custom scripts that may be implemented in a couple of ways.  Scripts can be used to create objects on reports at runtime or may change an existing XFF file. 

PRG based scripts

As the name implies, PRG based scripts are scripts defined in code within a PRG.  These can be either named scripts or page bound scripts.  There is no difference in the way you write a page bound script vs. a named script.  The difference is in how they are registered or called.  With a page bound script, you specify the pages on which the script is to be run when you register it.  A named script will only be run if a rectangle on a report calls it.
Rectangle based scripts

A rectangle based script is a script embedded in the comment field of a rectangle on a form.  What is embedded in this comment may either reference code that is located in a named script or it can have all the script code embedded directly in the comment of the rectangle.
Implementing page bound scripts

The XFRX documentation discusses implementation of the various kinds of scripts, so I won’t detail each one here.  I consider page bound scripts to be the most challenging, so that’s what I’ve used for the examples in my demo.
One important point is that the RegisterScript method must be called after you call the XFRX SetParams method. The manual only says, “Before XFRX is called…” which isn’t clear.

Look at the bottom of Main.prg and you’ll find a class definition called MyXFRXScripts.  The code for this class is shown below.

DEFINE CLASS MyXFRXScripts as Custom

   * Create a property for each script that is called the script name

   * plus "Pages".

   * This keeps the RegisterScript function from having to know which

   * pages we want the script to run on.

   * If there is no such property, RegisterXFRXScripts will treat it

   * as a named script.

   * Valid choices are:

   *
"ALL" for all pages

   *
"ODD" for odd pages

   *
"EVEN" for even pages

   *
"X" where X is a specific page number (note that this must be

   *        a string, not numeric)

   *
"X,Y,Z" where X, Y and Z are specific pages to print

   * 
"X-Z" where X is the starting page and Z is the ending page

   *
or you can mix the two choices above, e.g. "X,Y,Z,A-D"

   DrawWatermarkPages = "ALL"

   DrawBlueRectanglePages = "5"

   * If a script should only be registered under certain conditions, create a

   * property for the script name plus "Cond" and store the condition in that 

   * property.

   * This condition will be evaluated from the RegisterXFRXScripts method of

   * the DoReports class, so any references to "This" will be referring to

   * that class.

   DrawWatermarkCond = [inlist(this.cOutput,"PREVIEW","PDF")]

   *-----------------------------------------------------

   FUNCTION DrawWatermark

   * Draw a watermark

   LPARAMETERS oXFD

   oXFD.SetPos(100,400)

   oXFD.SetColor(210,210,210)

   oXFD.SetFont("Arial",100,.T.)

   * The second parameter of the DrawText method specifies the angle of

   * rotation in a counterclockwise direction.

   * Rotation will only appear for PDF output and when printing

   * from the preview.  Rotation in the preview will be implemented soon.
   oXFD.DrawText("Watermark",45)

   ENDFUNC 

   *-----------------------------------------------------

   PROCEDURE drawBlueRectangle

   * This example comes straight out of the XFRX manual

   LPARAMETERS oXFD

   oXFD.setColor(0,0,255,-1,-1,-1)

   oXFD.drawRectangle(100,200,50,50)

   ENDPROC

   *-----------------------------------------------------

ENDDEFINE

Within this class are two methods, one called DrawWatermark and one called DrawBlueRectangle.  The DrawBlueRectangle method was taken directly from the XFRX manual.

I’ve added properties to the class for each of the methods that store the pages on which the script should act.

To register a script in XFRX, instantiate the class containing your scripts and call the XFRX RegisterScript method, passing the name of the script, as shown below.

This.oScripts = CREATEOBJECT("MyXFRXScripts")

This.oRpt.RegisterScript(“DrawBlueRectangle”,0,”ALL”,0)

I would prefer not having to repeat code to call RegisterScript for each my scripts, and would prefer not even having to know the names of my scripts.  To achieve this, I created a method called RegisterXFRXScripts which uses VFP’s AMEMBERS function to get the method names.

FUNCTION RegisterXFRXScripts

LOCAL lcPages

* If any scripts were defined in MyXFRXScripts, find them and register them

This.oScripts = CREATEOBJECT("MyXFRXScripts")

* Find out what all the native members are so we can pick out the ones

* that aren't

AMEMBERS(aNative,This.oScripts,1,"N")

* Now get all the members

AMEMBERS(aScripts,This.oScripts,1)

FOR i = 1 TO ALEN(aScripts,1)

   IF UPPER(aScripts[i,2]) = "METHOD" AND ASCAN(aNative,aScripts[i,1]) = 0


* It's a method and it's not native


IF NOT PEMSTATUS(This.oScripts,aScripts[i,1]+"Cond",5) OR ;

                EVALUATE(EVALUATE('This.oScripts.'+aScripts[i,1]+"Cond"))


   IF PEMSTATUS(This.oScripts,aScripts[i,1]+"Pages",5)



* Register it as a page bound script



lcPages = EVALUATE('This.oScripts.'+aScripts[i,1]+"Pages")



* The third parameter specifies if the script output should be

            * under on top of report objects.  The fourth parameter specifies

            * the page numbering type--see XFRX manual.



This.oRpt.RegisterScript(This.oScripts,aScripts[i,1],0,lcPages,0)

 
   ELSE



* Register it as a named script



This.oRpt.RegisterScript(This.oScripts,aScripts[i,1])


   ENDIF 


ENDIF 

   ENDIF

ENDFOR 

ENDFUNC 

When you run the reports, you’ll see that each page has a large gray “Watermark” that appears on each page.  If you print all the reports, the last page (which is page 5) will have a blue rectangle on it.  See figure 3. These are the result of the scripts I have defined.

Note that there are two properties that can be defined in the Scripts class for each script that define the pages on which the script is to act and the condition under which the script should be registered. My RegisterXFRXScripts method check for these properties for each script within the class and will act accordingly.  
You must create a property for the pages if you want the script to be a page bound script.  Without the “Pages” property, the script will be registered as a named script.  

The condition property (“Cond”) is evaluated during the RegisterXFRXScripts method and will only register the script if the property doesn’t exist or evaluates to .T.  This can be handy in the case of the watermark that is supposed to be rotated at 45°.  Because the rotation only works with PDF and the preview (if you print), I only want to register the DrawWatermark script if you are sending output to a PDF file or the preview container.  I’ve created a DrawWatermarkCond property that contains code to check the output format to determine if the script should be registered.  Note that the code that evaluates this condition in the RegisterXFRXScript method uses a double EVALUATE() function.  This is because the first EVALUATE() function only returns the string store in the property.  The second EVALUATE() is what checks the condition.

Figure 3 – Report example showing results of running watermark and blue rectangle scripts
Introducing the Preview Container
XFRX comes with a very useful Preview Container.  Although I cover this in much more detail in my second session, I cover it just briefly here as well.  Figure 3 shows how the Preview Container appears in the Demo program.

Figure 3 – The Preview Container
All reports selected are included in a single preview, but unlike the Acrobat Reader, you can’t scroll to get from page to page.  The PageUp and PageDown buttons work for moving from page to page, as well as the navigation buttons on the toolbar. The toolbar also includes a button to allow you to jump to a specific page. 

The button on the far left of the toolbar will toggle whether or not the bookmarks are displayed.  

The “Search” button (with the binoculars icon) will show a dialog like that shown in Figure 4.

[image: image4.png]Ocurrent page.

Find Next





Figure 4 – The XFRX “Search” dialog
Scoping variables
If you call your reports from the Init of the preview form, the one issue you’re most likely to have problems with in using the container is the scoping of variables on your reports.  Using LOCAL variables or Thisform.SomeObject.Value works on reports called directly from a form, but this won’t work with reports called from the Preview Container.  The reason is that the Preview Container sits in its own form, so “Thisform” is no longer referring to the same form, and anything scoped to your original form is not available.  This problem is not hard to work around; it’s just something you need to be aware of.  If you use the Report Class to queue your reports, you can still use THISFORM, but LOCAL variables will be out of scope.
Using the container

I go into much more detail about the container in the second white paper.  You have access to that paper as well, so I won’t repeat most of that here.  There are only two things you need to be aware of in order to use the container:
1. Use the container class that comes with XFRX (xfcont) or use an instance of the preview form that comes with XFRX (XFRXfrmPreview). I recommend the latter.
2. You can either follow the example in the Init method of the Preview form that comes with XFRX versions before 11.1, or you can use my report class to queue up and run the reports for you before you call the preview form.  The preview form included with previous versions of XFRX shows a hard coded reports being called.  You will need a more flexible mechanism, which is what the Report Class provides.

More details about XFRX
XFRX has its own report engine which is used for VFP versions 8 and earlier.  For VFP 9, the native engine is used in conjunction with a special listener class.  This means that many of the commands used to run reports in VFP 9 are different from previous versions.  These are documented in the XFRX manual.
XFRX includes many features that I haven’t detailed before this point.  I discuss some of these below.

Mixing portrait and landscape reports

In order for XFRX to mix portrait and landscape reports in the same output, it must read the first record of each FRX.  In VFP 9, XFRX uses a report listener to do this, but it only works if the XFRX listener is the first in the chain.  This means if you need to use your own listener, you must add it to the chain AFTER the listener for XFRX.  You would do this as follows:

loMyListener = CREATEOBJECT("MyListenerClass")

loMyListener.IsSuccessor = .T.

loXFRXListener = xfrx("XFRX#LISTENER")

loXFRXListener.Successor = loMyListener

REPORT FORM (toRptObject.aRpts2Run[lnRpt,1]) OBJECT loXFRXListener
Bookmarks
XFRX allows you to use bookmarks for the Preview, PDF and HTML output formats.  You can easily turn bookmarks on and off at runtime.  I allow you to do this by setting a property in the Report Class called lXFRXBookMarks.  This property can be set for each report before calling the DoIt method of the Report Class.  To turn booksmarks on or off in XFRX directly, use the SetOtherParams(“PRINT_BOOKMARKS”,setting) command.
By default, bookmarks in HTML will include the bookmarks you’ve defined in your reports plus a bookmark for the top of each page.  This is not something that I particularly like, but version 11 of XFRX lets you change this by passing a third parameter to the SetOtherParams function. 

· Sending 0 will show both page number bookmarks and the ones you’ve defined in your reports (default)
· Sending 1 will show only the bookmarks you’ve defined in your reports
· Sending 2 will show only the page number bookmarks

To define bookmarks in a report, you need to add a comment to one of the fields with #UR OUTLINE= <expression> where <expression> returns the string to display for the bookmark text.  If you modify the Scores report included with the sample code, double-click on the Date() field in the upper left, and view the comments, you’ll see that I have  #UR OUTLINE="Score Sheet" in there.  This creates a bookmark that says “Score Sheet” for this report.  Because this is a one-page report, I don’t worry about the bookmark appearing on each page.  

If you have a multiple page report and you only want the bookmark to show for the first page, you need to add the comment to an object in the Title band of the report.  To see an example of this, modify the WIP report included with the sample code.  For this report, I don’t need a Title band for anything that should appear in the final output, other than the bookmark.  The way I got this to work is to create an object in the title band with a minimum height.  You can see this object because it prints white text on a white background.  Hold your mouse down in the left end of the Title band and drag it across the band.  You’ll see that an object is selected.  You can then double-click in this object and look at the comment to see the bookmark.
You may have a report with data groupings where you want a bookmark for the start of each of one of the groupings.  Use the same technique for adding a bookmark into one of the objects in the group header.

Bookmarks in the Preview Container

Everything I’ve described above applies to all output formats that can show bookmarks. The Preview Container has an additional feature. The ibook property of the container class specifies if the bookmark panel should be displayed when the preview is first displayed.  This property is undocumented, so I will tell you here what the settings are and what they do:

· -1 = Disable bookmarks

· 0 = Enable bookmarks but hide them (users need to click on the bookmark button to see them)

· 1 = Enable bookmarks, always showing them

· 2 = Enable bookmarks, but show or hide them automatically based on whether or not there are bookmarks

The SetParams method

If you use my Report Class, initialization of XFRX is hidden from you.  But you may want to modify the Report Class so it will allow output to some of the other formats supported by XFRX.  The InitXFRX method includes the code that confirms the output file and calls the SetParams method.  SetParams is where you tell XFRX which output format to use.  For example, to initialize XFRX for output to the container you would do the following:
oXFRX.SetParams(lcOutputFile,,,,.T.,,"XFF")

The “XFF” parameter tells XFRX to send output to an internal XFF file format.  See the XFRX documentation for details of how to call the SetParams method.
The XFRX Draw class
The Draw class is XFRX is where most of the work is performed.  In most situations, you don’t need to know much about this class.  If you are more adventurous, however, you can use this class to generate just about anything manually, including watermarks, hyperlinks and shapes. There is quite a bit of documentation about this class and I haven’t delved into using it other than what is in this demo, so I will leave this topic for you to explore further.
Distribution

Once you’ve completed your app and you’re ready to distribute it, you’ll need to include a few extra files so that XFRX will work with the EXE you distribute.  These files are:
· Hndlib.dll

· Xfrxlib.fll

· Zlib.dll

These files don’t have to be registered on the target machine.  Just copy them to the same directory as your app’s EXE file.

Troubleshooting

As with any third party tool, you may run into some problems when you try to integrate XFRX into your apps.  Here are some of the more likely problems you may have.

· An error message of “File xfrxlib.vcx does not exist” indicates that you’ve forgotten to SET PATH TO your XFRX and XFRX\Libs directories.

· If you don’t get any output, you may have forgotten to call the XFRX Finalize method.  If you are using my Report Class, call the SendRptToOutput method.
· If you get any errors referencing Zlib.dll, hndlib.dll or XFRXLib.fll when you try to run a report in production, then you’ve forgotten to copy one of the distribution file mentioned in the previous section to the main app directory.  None of these files need to be registered.

· When you don’t see something you’re expecting, it’s likely that an error was encountered and trapped by the preview container’s error method.

· A report that prints to the default printer no matter which printer you select, or doesn’t use other printer properties you’ve set is usually caused by printer information stored in the EXPR field of the FRX.  Make sure you clear any unnecessary information out of the EXPR field.  In VFP 8 and above, make sure you don’t have the “Printer environment” option on the Report menu pad checked.

· A landscape report displayed in portrait mode could be one of two problems:

· You don’t have the latest version of XFRX.  The initial release of version 11 for VFP 9 had a problem when mixing portrait and landscape formats in the same output.  Note that this problem could also cause a portrait report to appear in landscape mode in mixed output.
· Open your FRX and look at the EXPR field for the first record.  For landscape reports you must have:
ORIENTATION=1

PAPERSIZE=1
otherwise XFRX won’t show it as landscape.  Note that ORIENTATION=1 alone is NOT sufficient to create landscape output in this case!
· If your output is scrambled, like that shown in Figure 4, the problem is caused by using a new XFRXLib.fll with an older version of XFRX.fxp.

Figure 4 – Scrambled output
A word about documentation
Creating and maintaining thorough documentation for a product this robust can be a challenge for a single person, especially when that person is handling all development and support.  This is probably the reason that the documentation isn’t as complete as I would like it to be.  For example, there are many properties and methods for the Preview Container that are never mentioned in the documentation.  Most of these are probably things you would never care about, but some of them really should be discussed. The ShowStatus property is a good example of this.  By default, the Preview Container has toolbars at the top and bottom of the form.  I wanted to remove the toolbar at the bottom of the form.  I put in a request with Martin for this, and he let me know that the container already had a property called ShowStatus, and that setting it to .F. would hide the toolbar at the bottom.  As a testament to Martin, he has since added documentation for this property plus the iBook and iTool properties.
