
Overview

XFRX is a powerful third party tool for allowing you to send your FRX report files to various output formats.  The full version allows you to create reports in PDF, Word, HTML, Excel, RTF, XML, various kinds of image files, and other formats.  This session will focus on the advantages of using the XFRX preview container, as well as the quirks involved with its use.  We’ll pay particular attention to the challenges of printing from this preview container, including the display of a full properties dialog for whichever printer your user selects. This session will get into the use of the Windows API for controlling the printer and the use of the DEVMODE structure.
Why a Preview Container?
You may be asking, “Why use a third party preview, especially with the enhanced preview of VFP 9?”  Good question.  But VFP doesn’t do everything yet.  With XFRX you can preview multiple reports as if they were a single report.  You can also mix portrait and landscape reports in the same preview.  The XFRX preview container allows you to search for strings either on the current page or anywhere in the report.  XFRX also supports bookmarks and drill-down capabilities in its preview container.
At this point, you’re probably asking why not just output to PDF.  The free Acrobat reader will do all these things too.  This is another very good question. What if you’re creating a product and don’t want to have to deal with whether or not the reader is installed?  What if you’re in a situation where your client doesn’t want countless PDF files generated all over the hard disks?  Even if you intend to delete the PDF file when you’re done, that doesn’t stop the user from using the “Save As” feature to save a copy to a file your program doesn’t know about.  Also, the Reader has a button to send the file as e-mail right there on the toolbar.  My client didn’t want to make it so easy to e-mail reports for security reasons.  It was because of all of these reasons that I found myself needing to use the XFRX Preview Container.
Before we begin

What you need to know

In order to truly understand the examples for this session, you need to be familiar with the Report Class I use.  This is covered in my other session on Integrating XFRX into your VFP Applications, so you if you didn’t attend that session you should look over the white paper.

I strongly recommend you use the latest version of XFRX.  (As of this writing, that is version 11.1.)  Versions older than 10.2 don’t have the PrintDocument method, which allows you to print the internal XFF file that the preview generates.  I’ll discuss more on this and other issues later.

What you need to do

When you unzip the source code for this session, make sure you keep the relative path included with each file.  An evaluation version of XFRX is included, and the program expects it to be in a subdirectory called XFRX.

Mixing portrait and landscape reports

In order for XFRX to mix portrait and landscape reports in the same output, it must read the first record of each FRX.  In VFP 9, XFRX uses a report listener to do this, but it only works if the XFRX listener is the first in the chain.  This means if you need to use your own listener, you must add it to the chain AFTER the listener for XFRX.  You would do this as follows:

loMyListener = CREATEOBJECT("MyListenerClass")

loMyListener.IsSuccessor = .T.

loXFRXListener = xfrx("XFRX#LISTENER")

loXFRXListener.Successor = loMyListener

REPORT FORM (toRptObject.aRpts2Run[lnRpt,1]) OBJECT loXFRXListener
Preview Container basics

XFRX Classes to use
The XFRX Preview Container is a class called XFCont in the XFRXLib.vcx. Although the documentation for XFRX states that by adding a single line to the form’s Destroy method, you can drop the container class onto a any form, I’ve found this doesn’t quite work so well.  I much prefer to create an instance of the preview form class (XFRXfrmPreview in the XFRXLib.vcx).  This form also has code in the Show, Resize and Keypress methods that make the container work properly with the form.  I figured that if XFRX comes with the form class I need, why should I try to shoehorn the container into my own form?
So that you can see the difference in behavior, I’ve provided two forms in this demo that use the preview container.  One is called Preview.scx and is an instance of the XFRX preview form.  The other is called MyPreview.scx and is a native VFP form on which I’ve dropped an instance of the XFRX Preview Container, and have made the recommended addition to the Destroy method, plus the Init method I use in conjunction with my Report Class.  Look at the SendRptToOutput method of my Report Class and you’ll see that it’s currently calling only the Preview.scx.  To see the behavior of MyPreview.scx, comment out the call to Preview.scx and remove the comment for the call to MyPreview.scx.  Run the Demo program and send reports to the screen using XFRX.  Notice that although the form has maximized itself, the Preview Container has not, and you need to scroll around to see anything.

A more generic preview form

In versions of XFRX before 11.1, the preview demo form (the Preview.scx form in the XFRX/PrevDemo folder) has code in the Init method that demonstrates how to use the preview container in a form; however, the example uses hard coded report names.  I needed the preview form to be more generic than that, so I needed a scheme for generating reports that didn’t require hard coding.  I decided to add an array property to my Report Class that would store the name of the reports to be run, along with the alias that needed to be selected when the report was run.  This is the purpose of the aRpts2Run property.  I also moved the code that generates the reports into my Report Class so it’s centralized whether you are sending output to a file or the preview.
Scoping considerations

Another important point to be aware of with the use of the preview container is one of scoping.  The preview form uses the default datasession, so any tables and cursors that you opened or created for your report are available, but the reports are run outside of the form.  This means local variables are not available for your reports.  These should either be in the data used for the reports, or be in properties of an object that the Report Class has in scope.  You may want to add a property to the Report Class to hold an object reference to a parameter object.  
Modified Init method

In the instance of the XFRX Preview form I use, I have overridden the Init method with my own code.  One of the changes is that I’ve added a parameter which holds a reference to my Report Class, so all properties of that object are available.  Because I’ve called the REPORT FORM commands before calling the preview form, you will probably find the only property you need from my Report Class is the reference to the XFRX object.  The code below is my Init method.
LPARAMETERS toRptObject

local loObj

* Save the reference to the Report Class

* We need to send it to the PrintOptions form

thisform.oRptObject = toRptObject

* Code from Martin to get this to work in version 10.2

*-----------------------------------------------------

DECLARE Integer compress In zlib As cxx String @, String @, String @, ;

        Integer sourcelen

LOCAL la, lb, ld

la = "Test"

lb = space(500)

ld = space(500)

=cxx(@lb,@ld,@la,len(la))

IF ATC("xfrxlib.fll", SET("Library")) = 0

   SET LIBRARY TO xfrxlib.fll addi

ENDIF

*-----------------------------------------------------

IF toRptObject.lXFRXBookmarks

   * Show the bookmarks panel when the preview first opens
   This.CntXFRX.ibook = 2

ELSE

* Hide the bookmarks panel if we've turned off bookmarks

* in the report object
   This.cntXFRX.ibook = -1

ENDIF 

* This turns off the toolbar at the bottom of the screen

This.cntXFRX.ShowStatus = .F.


* Get a reference to the XFF file

thisform.oReport = toRptObject.oRpt.oXFDocument

thisform.cntXFRX.reset()

toRptObject.oRpt.XFRXPreviewer = thisform.cntXFRX

toRptObject.oRpt.PreviewReport(thisform.oReport)

toRptObject.oRpt.XFRXPreviewer = NULL

* Maximize the preview form

this.WindowState = 2

IF TYPE("This.cntXFRX.oTool")="O" AND NOT ISNULL(This.cntXFRX.oTool)

   * Dock the toolbar

   This.cntXFRX.oTool.dock(0)

ENDIF 

Printing from the Preview Container

Of course, my client who required that I use the Preview Container rather than PDF output was the same one who wanted to be able to print from the Preview Container.

Normally in VFP, when we preview a report and want the user to see a printer options dialog when clicking on the Print button from the preview’s toolbar, we would call the report with the REPORT FORM command, adding TO PRINTER PROMPT PREVIEW clauses.  But by using the XFRX Preview Container, I wasn’t using the VFP preview toolbar, so the Print button didn’t work that way.

At the time I was given this task, XFRX had no PrintDocument method, nor code in the print button on the toolbar, nor a Print method of the container.

My client wanted to be able to select specific pages to print.  This meant that I couldn’t use GETPRINTER() or SYS(1037) because neither of those dialogs allows you to specify pages to print.  The only Windows dialog that allows you to select pages is the one used when you call a report with TO PRINTER PROMPT PREVIEW.   I realized that if my client was going to have an option to select pages it meant I’d have to create my own form that looked like a Windows dialog.  Figure 1 shows the dialog I created, but the original version of this dialog didn’t have the “Properties” button.

Figure 1 – My custom printer dialog
I was successful in removing the Print button that instantiated on the toolbar in the Preview Container, substituting it with my own that called my new form.  My next challenge was how to actually do the printing.
Attempting to use VFP’s native engine to print
The user’s selection of which pages to print was based on the continuous page numbering shown in the preview.  In order to figure out which reports contained the requested pages, I had to run the reports twice—once to create a “page map” of which reports contained which starting and ending pages in the continuous page numbering sequence—then a second time to actually print out the pages.  This also required that I use a variable that stayed in scope between reports to keep track of pages numbers, and that I use this variable to show the page number on the reports so that the printed reports also had continuous page numbering.

After creating the page map and the code necessary to print the selected pages, I discovered that the XFRX Preview didn’t always have page breaks in the same place as the native VFP engine.  So the pages the user selected weren’t necessarily printing the same information that displayed in the preview.  This was a problem I couldn’t solve using the native engine!

Saved by the PrintDocument method

Fortunately, right about that time, Martin was working on version 10.2 of XFRX which has the PrintDocument method.  There was still no Print method of the container in the early version I had, so I was still using my own print button to call up my own printer dialog, but at least I could use the XFRX PrintDocument method and know that it would print what was shown in the preview.  The PrintDocument method takes four parameters:
1. The name of the printer

2. The name of the print job

3. The starting page number

4. The ending page number

When rendering reports for the Preview Container, XFRX saves the output in an internal file format that has an XFF extension.  XFRX keeps a reference to this file and the PrintDocument method will print the contents of the current XFF file.
You may want to skip the next few sections if you just want to get to the solution.  These sections cover a lot of the learning curve I went through to get to my solution, but you don’t need to know most of this to use the solution I came up with.  See the section titled, “The ultimate solution – new additions to XFRX”.

The DeviceCapabilities API
Wouldn’t you know it—the next thing the client complained about was the fact that they couldn’t select which paper tray to use when selecting the printer.

I started asking around.  At first I was given a routine that would retrieve the trays available for each printer, so I could at least have a dropdown for selecting paper trays.  The code to do this is in the PrinterTrays.prg that comes with this session’s source code, and is shown below.
LOCAL lnIndex, lcPrinterName, lcPort, lcBinNameBfr, lcBinIdBfr, lnTrayCnt, lnTrayCnt2, lnTrayIndex, lcTrayName, lnTrayID

WAIT WINDOW NOWAIT "Finding printer trays...please wait"

#DEFINE DC_BINS       6

#DEFINE DC_BINNAMES   12

DECLARE INTEGER DeviceCapabilities IN winspool.drv;

  STRING pDevice, STRING pPort, INTEGER fwCapability,;

  STRING @pOutput, INTEGER pDevMode

CREATE CURSOR PrinterTrays (TrayName C (25), TrayID I, PrinterPos I)

PUBLIC ARRAY aAvailPrinters[1]

APRINTERS(aAvailPrinters)

* Loop through all of the printers and get tray info.

FOR lnIndex = 1 TO ALEN(aAvailPrinters,1)

   lcPrinterName = aAvailPrinters(lnIndex,1)

   lcPort = ALLTRIM(aAvailPrinters(lnIndex,2))

   lcBinNameBfr   = REPLICATE(CHR(0), 16384)

   lcBinIdBfr   = REPLICATE(CHR(0), 1024)

   lnTrayCnt  = DeviceCapabilities(lcPrinterName, lcPort, DC_BINNAMES, ;

                @lcBinNameBfr, 0)

   lnTrayCnt2 = DeviceCapabilities(lcPrinterName, lcPort, DC_BINS, ;

                @lcBinIdBfr, 0)

   IF lnTrayCnt > 0


IF lnTrayCnt <> lnTrayCnt2


   MESSAGEBOX("Unable to match IDs to Tray Names for Printer " ;

                 +aAvailprinters[lnIndex,1],48,'Problem with printer trays')


ELSE 


   * Each string buffer is 24 characters long, terminated with

         * a NULL / chr(0) and contains the name of a paper bin

  
   FOR lnTrayIndex=1 TO lnTrayCnt


      lcTrayName = SUBSTR(lcBinNameBfr,(lnTrayIndex-1)*24+1,24)+CHR(0)


      lcTrayName = SUBSTR(lcTrayName, 1, AT(CHR(0), lcTrayName)-1)



lnTrayId = CTOBIN(SUBSTR(lcBinIdBfr, (lnTrayIndex*2)-1, 2),'2sr')



INSERT INTO PrinterTrays (PrinterPos,TrayName,TrayID) VALUES ;

                                     (lnIndex,lcTrayName,lnTrayId)


   ENDFOR


ENDIF && lnTrayCnt <> lnTrayCnt2

   ENDIF && lnTrayCnt <> 0

ENDFOR

WAIT CLEAR 

The PrinterTrays cursor holds all the available trays for each printer installed. The printer that corresponds to each available tray can be determined by the PrinterPos field, which matches the row for that printer in the aAvailPrinters array.  Although I didn’t make this array Public in the code I was really using, I did for this example so that you can run the code from the command window and compare the contents of the array with the PrinterTrays cursor.

Using this code, I could pull the contents for all the records that matched the current printer and use that to populate a combo, allowing the user to select a tray.  The problem was that although this code allowed selection of the available printer trays, it didn’t change the tray to be used.   Also, I knew it wouldn’t be long before the client wanted to change some other printer property.  Even the use of the dropdown for the trays was only a temporary solution.

DEVMODE structure and calling API functions

Then, Boudewijn Lutgerink offered some code on www.Foxite.com that he found on www.News2News.com which showed an example of setting printer properties and used the DocumentProperties API function to display a real Windows printer properties dialog for the current printer.  The code Boudewijn gave me is shown below.

Configuring DEVMODE structure for a printer

* Note: All these constants except for offsets can be found in the Windows.h file

#DEFINE IDOK     1
#DEFINE IDCANCEL 2

#DEFINE DM_OUT_BUFFER 2
#DEFINE DM_IN_BUFFER  8
#DEFINE DM_IN_PROMPT  4

#DEFINE DM_ORIENTATION  1
#DEFINE DM_PAPERSIZE    2
#DEFINE DM_PAPERLENGTH  4
#DEFINE DM_PAPERWIDTH   8

#DEFINE DMORIENT_PORTRAIT  1
#DEFINE DMORIENT_LANDSCAPE 2
#DEFINE OFFS_ORIENT_FLAG   41  && offset to Orientation flag
#DEFINE OFFS_ORIENT_VALUE  45  && offset to Orientation value

DO APICalls

PRIVATE hWindow, cPrinter, cDriver, cPort, hPrinter, nBufsize, cDevmode
hWindow = GetActiveWindow()
STORE "" TO cPrinter, cDriver, cPort
DO GetPrnName

hPrinter = 0
OpenPrinter(cPrinter, @hPrinter, 0)
IF hPrinter = 0
   Messagebox("Could not open default printer.", 48, "Error")
   RETURN
ENDIF

* being called this way the function returns DEVMODE buffer required size
nBufsize = DocumentProperties(hWindow, hPrinter, cPrinter,0,0,0)
cDevmode = Repli(Chr(0), nBufsize)

DO SetOrient WITH DMORIENT_LANDSCAPE
DO OpenDialog

ClosePrinter(hPrinter)
* end of main

PROCEDURE SetOrient(nOrient)
   * DEVMODE is adjusted programmatically
   * DM_IN_PROMPT not used -- the dialog window is not shown
   LOCAL dmFlags

   * reading printer configuration
   DocumentProperties(hWindow, hPrinter, cPrinter, @cDevmode, 0, ;

                      DM_OUT_BUFFER)

   * setting ORIENTATION flag and value
   dmFlags = buf2word(SUBSTR(cDevmode, OFFS_ORIENT_FLAG, 4))
   dmFlags = BitOr(dmFlags, DM_ORIENTATION)
   cDevmode = STUFF(cDevmode, OFFS_ORIENT_FLAG, 4, num2dword(dmFlags))
   cDevmode = STUFF(cDevmode, OFFS_ORIENT_VALUE, 2, num2word(nOrient))

   * using modified DEVMODE structure in the next call
   DocumentProperties(hWindow, hPrinter, cPrinter, @cDevmode, @cDevmode, DM_IN_BUFFER+DM_OUT_BUFFER)
ENDPROC

PROCEDURE OpenDialog
* showing printer settings dialog
   DocumentProperties(hWindow, hPrinter, cPrinter, @cDevmode, @cDevmode, DM_IN_PROMPT+DM_IN_BUFFER+DM_OUT_BUFFER)
ENDPROC

PROCEDURE GetPrnName
* retrieves default printer name from the Registry
   LOCAL lcBuffer, lcPrinter, lcDriver, lcPort

   lcBuffer = Repli(Chr(0), 120)
   GetProfileString("Windows", "Device", ",,,",   @lcBuffer, Len(lcBuffer))

   lcBuffer = STRTRAN(lcBuffer, Chr(0), "")

   DIMENSION arrPos[3]
   arrPos[1] = AT(",", lcBuffer, 1)
   arrPos[2] = AT(",", lcBuffer, 2)
   arrPos[3] = Len(lcBuffer)

   cPrinter = SUBSTR(lcBuffer, 1, arrPos[1]-1)
   cDriver = SUBSTR(lcBuffer, arrPos[1]+1, arrPos[2]-arrPos[1]-1)
   cPort = SUBSTR(lcBuffer, arrPos[2]+1, arrPos[3]-arrPos[2])
   RELEASE arrPos
ENDPROC

FUNCTION buf2dword (lcBuffer)
   RETURN Asc(SUBSTR(lcBuffer, 1,1)) + ;
   BitLShift(Asc(SUBSTR(lcBuffer, 2,1)),  8) +;
   BitLShift(Asc(SUBSTR(lcBuffer, 3,1)), 16) +;
   BitLShift(Asc(SUBSTR(lcBuffer, 4,1)), 24)
ENDFUNC

FUNCTION buf2word (lcBuffer)
   RETURN Asc(SUBSTR(lcBuffer, 1,1)) + ;
          Asc(SUBSTR(lcBuffer, 2,1)) * 256
ENDFUNC

FUNCTION num2dword (lnValue)
   #DEFINE m0       256
   #DEFINE m1     65536
   #DEFINE m2  16777216
   LOCAL b0, b1, b2, b3
   b3 = Int(lnValue/m2)
   b2 = Int((lnValue - b3*m2)/m1)
   b1 = Int((lnValue - b3*m2 - b2*m1)/m0)
   b0 = Mod(lnValue, m0)
   RETURN Chr(b0)+Chr(b1)+Chr(b2)+Chr(b3)
ENDFUNC

FUNCTION num2word(lnValue)
   RETURN Chr(MOD(m.lnValue,256)) + CHR(INT(m.lnValue/256))
ENDFUNC

PROCEDURE APICalls
   DECLARE INTEGER OpenPrinter IN winspool.drv;
      STRING pPrinterName, INTEGER @phPrinter, INTEGER pDefault

   DECLARE INTEGER GetProfileString IN kernel32;
      STRING lpApp, STRING lpKey, STRING lpDefault,;
      STRING @lpReturnedString, INTEGER nSize

   DECLARE INTEGER ClosePrinter IN winspool.drv INTEGER hPrinter 

   DECLARE INTEGER GetActiveWindow IN user32

   DECLARE INTEGER DeleteDC IN gdi32 INTEGER hdc

   DECLARE INTEGER DocumentProperties IN winspool.drv;
      INTEGER hWnd, INTEGER hPrinter, STRING pDeviceName,;
      STRING @pDevModeOutput, STRING @pDevModeInput, INTEGER fMode


ENDPROC
With some brief testing, I found this dialog really was the dialog for the specific printer specified, and would change if the printer changed.  I thought I had struck gold!

Learning about DEVMODE

One thing that using the DocumentProperties API function required was that I become familiar with the DEVMODE structure, which is integral to using the function. Just about all the documentation I could find for DocumentProperties and the DEVMODE structure included examples in C.  Some were in VB.  But NONE were in VFP.  (What a surprise!)  You can still figure out what the DEVMODE structure is about by reading these sites:

How To Modify Printer Settings with the DocumentProperties() Function

http://support.microsoft.com/default.aspx?scid=kb;EN-US;Q167345 

DEVMODE Structure
http://msdn.microsoft.com/library/default.asp?url=/library/en-us/vcmfc98/html/_mfc_devmode_structure.asp 
The DEVMODE structure consists of two parts, the public part, and the private part.  If you only care about the public part of the structure, you can write code that will apply to any printer.  Some printers will also have a private section that they add to DEVMODE, and you must consult the documentation for that printer in order to know how to use that section.  Here is what the common (public) part of the DEVMODE structure looks like:

    TCHAR  dmDeviceName[32];

    WORD   dmSpecVersion;

    WORD   dmDriverVersion;

    WORD   dmSize;

    WORD   dmDriverExtra;

    DWORD  dmFields;

    short  dmOrientation;

    short  dmPaperSize;

    short  dmPaperLength;

    short  dmPaperWidth;

    short  dmScale;

    short  dmCopies;

    short  dmDefaultSource;

    short  dmPrintQuality;

    short  dmColor;

    short  dmDuplex;

    short  dmYResolution;

    short  dmTTOption;

    short  dmCollate;

    TCHAR  dmFormName[32];

    WORD   dmUnusedPadding;

    USHORT dmBitsPerPel;

    DWORD  dmPelsWidth;

    DWORD  dmPelsHeight;

    DWORD  dmDisplayFlags;

    DWORD  dmDisplayFrequency;
Knowing the structure for DEVMODE, look back at the code from News2News.  The cDevmode variable in the code stores the DEVMODE structure as a binary string.  The functions buf2dword, buf2word, num2dword and num2word are used to translate to and from binary.  However, in VFP it would make more sense to use the CTOBIN() and BINTOC() functions instead.
Word types are 2 bytes (16 bits) and DWord types (double word) are 4 bytes (32 bits). Short is 2 bytes. Using this knowledge, you can see how the code posted on the News2News site derived the offsets for the Orientation flag (41) and the Orientation setting (45).  I think that the variable name for the orientation flag is misleading.  This is the offset to the entire dmFields.  dmFields stores flags for each of the settings below that shows whether or not each setting is available for the current printer.  The position of each bit in dmFields corresponds to its position in the structure. (Or so the documentation says.)  Orientation is the first property, which is why the DM_ORIENTATION constant from the Windows.H file is 1.  The first byte of dmFields is the 41st byte of the structure.  dmFields is 4 bytes long, so the setting for Orientation starts at the 45th byte of the structure.

Using similar logic, I’ve figured out the offsets for the duplex setting.  It works out that the offset for the duplex setting is 63.

Remember that dmFields has a bit for each property.  The Windows.h file has a constant set to the decimal value for each of these flags when the bit is turned on.  DM_ORIENTATION is 1, DM_PAPERSIZE is 2, DM_PAPERLENGTH is 4, etc.  I’ve included the Windows.h file with the source code for this session so you can see this easily.  But if you look carefully at the constants, you’ll see that the values of 32, 64, and 128 are missing between DM_SCALE and DM_COPIES.  I can only assume that Microsoft decided not to use these bits in dmFields for some reason.

Expanding the code example

Looking at the code from News2News more closely, as well as examples of similar functions in C, I noticed some problems with the code from News2News.  You must check dmFields to see if the flag is turned on for the particular property you intend to change to determine if it’s available for the current printer.  And you should not change anything in dmFields.  The code from News2News was calling BITOR rather than BITAND, which isn’t going to tell you whether or not the bit is set.
  Besides, it’s not checking the return value.  And it’s changing dmFields!  I’ve changed this part of the code as follows:
dmFlags = buf2word(SUBSTR(cDevmode, OFFS_ORIENT_FLAG, 4))

IF BITAND(dmFlags, DM_ORIENTATION) <> 0
  cDevmode = STUFF(cDevmode, OFFS_ORIENT_VALUE, 2, num2word(nOrient))

ENDIF

After working with the code from News2News for a brief time, I realized I didn’t need the GetPrnName function, which only returns the name of the default Windows printer; I could just pass the printer name to this code as a parameter from my own form.  But another problem that became clear was that this function doesn’t actually send any changes back to the printer.  It only hard codes the orientation to landscape, and then displays a properties dialog so you can see the change.  But when I tried to send a report to the printer, it wasn’t in landscape, even if the Expr, Tag and Tag2 fields of the FRX were empty!  Even worse, this technique involved manually coding every possible property, which I didn’t want to do!  I just wanted something that would put up a property sheet and send all the settings, whatever they may be to the printer.

I started asking around on the UniversalThread and Foxite.com, and got some good feedback, but nothing that solved the problem of writing the changes to the printer so that VFP would use the settings.  A couple of people even insisted that it wasn’t possible in VFP without an FLL or DLL file created in C.  This may be true for VFP 8 and earlier, but it’s not true for VFP 9.

I did come up with a way to program individual properties and send those to the printer. If your goal is to control printer properties programmatically, see Appendix A.  There you’ll see my solution for controlling the printer programmatically.  At this point, though, I’ll switch to discussion of dealing with the entire set of properties as a unit.
SYS(1037) to the rescue

Finally, I decided to talk to some folks on the VFP team at Microsoft about my problem.  Both John Kozial and Richard Stanton suggested using the SYS(1037) function with the new enhancements for VFP 9.

In VFP 9, a second parameter has been added to SYS(1037) which allows it to work in conjunction with an FRX for storing and writing out a DEVMODE structure.  Send 0 or omitting the second parameter makes SYS(1037) work as it has in previous versions.  The other possible values for this parameter require that your current work area have an FRX opened exclusively.  The part that I didn’t understand right away was that this FRX doesn’t have to belong to any report you’re trying to run.  The FRX is only used by SYS(1037) as a vehicle for reading and writing out the DEVMODE structure.

Assuming you have an FRX opened exclusively, the second parameter works as follows:

· 1 will open the report Page Setup dialog.  This dialog is identical to the one you get without the second parameter, but sending 1 as the second parameter saves any changes you make into the FRX file.  If you don’t have an FRX open, SYS(1037,1) doesn’t display the dialog.
· 2 reads the current printer settings and loads them into the FRX

· 3 sends the DEVMODE structure stored in the TAG2 field to the current printer.  If you don’t have complete information in the FRX, i.e. the EXPR, TAG or TAG2 fields aren’t filled in, SYS(1037,3) stores the current information into those fields as well.
This last one is probably confusing to you.  It was to me.  According to the help file, “SYS(1037,3) restores printer settings from a report definition to the current Visual FoxPro default. Although it may seem counter-intuitive, using this variant of the function requires complete printer environment information in the report or label, and therefore may write data to the cursor as well as changing the default printer environment. If your report or label did not have complete information before you used the function, it will have complete information after you use it.”  What this means is if you have a DEVMODE string stored in the TAG2 field, that is what will be sent to the printer.  If you call SYS(1037,3) and you have an empty TAG2 field the current settings will be saved to the FRX.  Apparently, if you make any changes in a printer properties dialog, those are considered current settings and are saved to both the printer and the FRX file.
Here are the steps to use SYS(1037) to load and set printer settings:

1. Use your  temporary frx exclusive.

2. Use SYS(1037) to show the printer dialog and let the user change settings.  This call will return “1” if changes were made.

3. Blank out the Expr, Tag and Tag2 fields so that anything in there doesn’t interfere with what you will load.
4. Use SYS(1037,3) to send the DEVMODE structure out to the current printer.  This will also update the EXPR, TAG and TAG2 fields of the FRX with the current settings.  

5. The TAG2 field holds the new DEVMODE structure. Save the contents of the TAG2 field for later use.
Another structure called DEVNAMES is stored in the Tag field, but we don’t need to get into what that’s used for.  I’ve mentioned it here in case you’re curious and would like to explore further.

Using both DocumentProperties and SYS(1037)

After implementing the SYS(1037) solution the first thing I noticed was that it was cumbersome in conjunction with the form I had already created. Figure 2 shows my imitation Windows printer options form.  The three buttons under the “Name” dropdown are unusual, and normally, you would have a single button that says, “Properties” like in Figure 1, but other than that, this screen looks very much like a standard Windows screen.  Note that you can select the printer on this screen and would only click on the one of the buttons below the printer selection in cases where you need to set additional printer properties.

[image: image2.png]Auto HP LaserJet 5P on INET





Figure 2 – My extended form with three properties buttons
Using the SYS(1037) solution, you would click on a “Properties” button to set additional properties. Figure 3 shows the properties dialog we get when we use SYS(1037)

[image: image3.png]



Figure 3 – First dialog with SYS(1037)
This is fine if all we want to do is change the paper tray.  But what if want to change something else like the duplex setting?  We’d have to click on the “Printer…” button, which shows the dialog in Figure 4.

Figure 4 – Second dialog with SYS(1037)
This seems redundant, since we selected the printer before we started, but we must once again click on the “Properties…” button, which finally gets us to a dialog where we can select the duplex mode.  See Figure 5.  And of course, this also means we have to click OK on all three dialogs to get back to the first dialog.

Figure 5 – Printer specific, third dialog with SYS(1037)
After pondering this situation for a while, I realized that the code I had been using with the DocumentProperties API was behaving the way I wanted to as far as the dialogs shown—it went straight to the dialog in Figure 5.  I just couldn’t use DocumentProperties to read or send the DEVMODE to the printer in such a way that VFP would use it.  But I knew that SYS(1037,3) would.  So I came up with a solution that uses DocumentProperties to show the dialogs and set a string with the DEVMODE structure, and then saves that DEVMODE structure in a temporary FRX so that I can use SYS(1037,3) just to write the changes out to the printer.  The result is that when you click on the “DocumentProperties” button from the first dialog, you go directly to the last dialog in Figure 5, where you can set any printer properties you want.

This solution is certainly more long-winded than the one that uses SYS(1037) alone.  Compare the two techniques.  First, the technique that uses SYS(1037), which can be found in the PrinterPropertiesSys method of the PrintOptions form:

* Lets the user set all possible printer properties

LOCAL lcRptFile

* Use a unique file name so we can use this in a multi-user situation

* Using a cursor instead of a physical file doesn't work, but we can

* create the FRX from a cursor.

lcRptFile = SYS(3)+".FRX"

CREATE CURSOR TempCur (Temp C (10))

CREATE REPORT (JUSTSTEM(lcRptFile)) FROM TempCur

USE IN TempCur

USE (lcRptFile) EXCLUSIVE ALIAS RptFile

IF EMPTY(Thisform.cOldExpr)

   * We only want to save the original settings the first time

   SYS(1037,2)

&& Read the current defaults

   Thisform.cOldExpr = EXPR

   Thisform.cOldTag  = TAG

   Thisform.cOldTag2 = TAG2

ENDIF 

* Display the dialog

IF SYS(1037) = "1"
&& Something was changed

   replace expr WITH '', ;

           tag WITH '', ;

           tag2 WITH ''

   * Writes the printer settings out to the printer and the FRX

   SYS(1037,3)


   Thisform.cDevmode = Tag2

ENDIF 

* Get rid of the temporary FRX/FRT

USE IN RptFile

ERASE (JUSTSTEM(lcRptFile)+'.*')

Fairly short and sweet, isn’t it?  Now let’s take a look at what’s necessary to use DocumentProperties in combination with SYS(1037).  This code can be found in the PrinterPropertiesDP method of the PrintOptions form:

* Lets the user set all possible printer properties

LOCAL lcRptFile, lhWindow, lcOrigDevMode, lcModifiedDevMode, lcPrinter, lhPrinter

* These constants come from the Windows.h file

#DEFINE IDOK     1

#DEFINE IDCANCEL 2

#DEFINE DM_OUT_BUFFER 2

#DEFINE DM_IN_BUFFER  8

#DEFINE DM_IN_PROMPT  4

DECLARE INTEGER OpenPrinter IN winspool.drv ;

      STRING pPrinterName, ;

      INTEGER @phPrinter, ;

      INTEGER pDefault

DECLARE INTEGER GetActiveWindow IN user32

DECLARE INTEGER DocumentProperties IN winspool.drv ;

      INTEGER hWnd, ;

      INTEGER hPrinter, ;

      STRING pDeviceName, ;

      STRING @pDevModeOutput, ;

      STRING @pDevModeInput, ;

      INTEGER fMode

DECLARE INTEGER ClosePrinter IN winspool.drv INTEGER hPrinter 

lcPrinter = ALLTRIM(thisform.cboName.Value)

IF NOT EMPTY(lcPrinter)

   lhWindow = GetActiveWindow()

   * If we have a printer open from a previous call, close it so we don't

   * keep extra handles open.

   IF NOT ISNULL(Thisform.hPrinter) AND NOT EMPTY(Thisform.hPrinter)


lhPrinter = thisform.hPrinter


ClosePrinter(lhPrinter)

   ENDIF 

   lhPrinter = 0

   OpenPrinter(lcPrinter, @lhPrinter, 0)

   IF lhPrinter = 0


Messagebox("Could not open printer.", 48, "Error")


RETURN

   ELSE


Thisform.hPrinter = lhPrinter

   ENDIF

   * Use a unique file name so we can use this in a multi-user situation

   * Using a cursor instead of a physical file doesn't work, but we can

   * create the FRX from a cursor.

   lcRptFile = SYS(3)+".FRX"

   CREATE CURSOR TempCur (Temp C (10))

   CREATE REPORT (JUSTSTEM(lcRptFile)) FROM TempCur

   USE IN TempCur

   USE (lcRptFile) EXCLUSIVE ALIAS RptFile

   IF EMPTY(Thisform.cOldExpr)

   * Use SYS(1037,2) to read the printer settings instead

   * of DocumentProperties


SYS(1037,2)


* We only want to save the original settings the first time


* Use a unique file name so we can use this in a multi-user situation


* Using a cursor here instead of a file doesn't work


Thisform.cOldExpr = EXPR


Thisform.cOldTag  = TAG


Thisform.cOldTag2 = TAG2


lcOrigDevmode     = TAG2


lcModifiedDevmode = TAG2

   ELSE 


lcOrigDevmode     = Thisform.cDevmode


lcModifiedDevmode = Thisform.cDevmode

   ENDIF 

   * Show printer settings dialog.

   lnResult = DocumentProperties(lhWindow, lhPrinter, lcPrinter, ;

              @lcModifiedDevmode, @lcOrigDevmode, ; 

              DM_IN_PROMPT+DM_IN_BUFFER+DM_OUT_BUFFER)

   IF lnResult <> IDCANCEL


* Set the printer to the new options


SELECT RptFile


replace expr WITH '', ;



  tag WITH '', ;



  tag2 WITH lcModifiedDevmode


thisform.cDevMode = lcModifiedDevmode


SYS(1037,3)

&& Writes the printer settings out to the printer

   ENDIF 

   * Get rid of the temporary FRX

   USE IN RptFile

   ERASE (JUSTSTEM(lcRptFile)+".*")

ENDIF 

This is definitely more complex, but I think you’ll agree that the results are much more acceptable.
The ultimate solution – new additions to XFRX

Version 11.1 of XFRX has a new _xfPrinterProperties method that will show a full properties dialog for any printer.  It’s very similar to using the DocumentProperties approach, but it’s much easier for you to implement.  The PrinterPropertiesXFRX method of the PrintOptions form shows the use of this technique.

* Lets the user set all possible printer properties

LOCAL lcRptFile, lcDevmode, lcNewDevmode
* Use SYS(1037,2) to read the current settings so that's what's

* displayed when we call _xfPrinterProperties()

* Use a unique file name so we can use this in a multi-user situation

* Using a cursor instead of a physical file doesn't work, but we can

* create the FRX from a cursor.

lcRptFile = SYS(3)+".FRX"

CREATE CURSOR TempCur (Temp C (10))

CREATE REPORT (JUSTSTEM(lcRptFile)) FROM TempCur

USE IN TempCur

USE (lcRptFile) EXCLUSIVE ALIAS RptFile

replace expr WITH '', ;


  tag WITH '', ;


  tag2 WITH ''

SYS(1037,2)

&& Read the current defaults

IF EMPTY(Thisform.cOldExpr)

   * Only save the first time

   Thisform.cOldExpr = EXPR

   Thisform.cOldTag  = TAG

   Thisform.cOldTag2 = TAG2

   lcDevmode = Tag2

ELSE

   lcDevmode = Thisform.cDevmode

ENDIF 

* _xfPrinterProperties parameters:

*  1 - Printer name

*  2 - String for DevMode

*  3 - Should the dialog show?

lcNewDevmode = _xfPrinterProperties(ALLTRIM(thisform.cboName.Value), ; 

                                    lcDevmode, .T.)

IF NOT EMPTY(lcNewDevmode)

   * If the user clicked Cancel from the dialog, we want to leave 

   * the properties as they were.

   Thisform.cDevmode = lcNewDevmode

ENDIF 

* Get rid of the temporary FRX/FRT

USE IN RptFile

ERASE (JUSTSTEM(lcRptFile)+'.*')

Note that I’m still using SYS(1037,2) to read the current printer settings.  This is only so that I can restore the settings when I close the form.  I also am making sure that I initialize lcDevmode to whatever the current settings are so that XFRX will display the current settings properly in the dialog.  Without all that, the only lines of code I would need for this technique are
lcNewDevmode = _xfPrinterProperties(ALLTRIM(thisform.cboName.Value), ; 

                                    lcDevmode, .T.)

IF NOT EMPTY(lcNewDevmode)

   * If the user clicked Cancel from the dialog, we want to leave 

   * the properties as they were.

   Thisform.cDevmode = lcNewDevmode

ENDIF
The _xfPrinterProperties method will return the following:
· The new DEVMODE structure if the dialog was displayed and the user clicked OK

· An empty string if the dialog was displayed and the user clicked Cancel

· The default DEVMODE structure if the dialog wasn’t displayed

Because I’ve specified that the dialog should be displayed (I sent .T. for the third parameter), I’m not concerned with the third possibility in my code.

Scope of changes
The scope of the settings from using SYS(1037,3) is local to the current VFP session.  A SET PRINTER TO command will usually cause the changes to revert to the defaults as well, although this doesn’t necessarily seem to work the DocumentProperties technique.  To make sure that I cleanup after myself on the way out of the form, I call a cleanup method, which loads the original values into a temporary FRX and writes those settings out to the printer.
Your choice of dialogs
When Martin released version 10.2 of XFRX it included the Print method of the Preview Container and a simple form as the default printer options dialog in which the user can specify the printer and specific pages.  See Figure 6.  This meant I didn’t have to use my own button class on the preview’s toolbar anymore, which was very appealing.

Figure 6 – The XFRX printer options dialog
With version 11.1 Martin added the “Printer properties” button.  Without that, this was a fairly simple dialog and it was likely you’d want to substitute one that offers more options.  Now with the “Printer proerties” button, you have access to the complete settings for the printer.  You can still use your own code in place of this form if you want to. To do this, you simply override the Print method of the container object (cntXFRX in the Preview form of this demo).  I have done just that in this demo.  If you look at the code of the Print method, you’ll see the following code:

LPARAMETERS cText

LOCAL lnWhichForm

DO FORM WhichForm TO lnWhichForm

DO CASE 

CASE lnWhichForm = 1

   DODEFAULT()

CASE lnWhichForm = 2

   DO FORM PrintOptions WITH thisform.oRptObject, this

ENDCASE
For this demo, I’ve created a form called WhichForm that asks if you want to see the default print dialog that comes with XFRX or if you want to use the Extended Printer Dialog form that I’ve created. See Figure 7.
[image: image7.png]Selecprinter0ptbns o

Which Printer Options Form to Use?

©(XFRX Default Form}
© My Extended Form

OK

[ cancel





Figure 7 – Custom form called from Print method
As you can see, by using the Print method as a hook for your own code you have quite a bit of flexibility.
Print it!

Now that we know the properties settings that the user wants, we can do the actual printing.  The PrintDocument method of XFRX allows us to print from the XFF file we generated.  PrintDocument takes up to five parameters:

1. The name of the printer to use

2. The name of the print job

3. The starting page number

4. The ending page number

5. A string containing the DEVMODE structure

Note that if you want to send both starting and ending page numbers, the third and fourth parameters must be numeric.  However, you can send a string as the third parameter that has page number ranges expressed as a comma delimited list, a range with dashes, or a combination of both like this “1,2,5-6,8,20-30”.  Although the form I’ve created restricts the entry of selected pages to a single page range, you can implement a more complex mechanism if you’d like, and XFRX will handle it.
The fifth parameter is new in version 11.1.  This parameter allows you to send a DEVMODE structure to the PrintDocument method so that it will use those settings.  Note that even though you may have written these settings to the printer, the PrintDocument method will use the default settings if you don’t include this parameter.

The PrtRpts method of the PrtOut form shows how I’ve implemented printing using the XFRX PrintDocument method in conjunction with my properties dialog:
LOCAL lnCopies, lnPrintRange, lnPagesFrom, lnPagesTo, lnStartingRpt, loXFF, ;

      lnCollate, lnCurPage, lcPrinter, lcDevmode

WAIT WINDOW NOWAIT 'Printing report...'

lnCopies = thisform.spnCopies.Value 

lnPrintRange = thisform.opgPrintRange.Value

lnPagesFrom = thisform.txtPagesFrom.Value

lnPagesTo = thisform.txtPagesTo.Value

lnCollate = thisform.chkCollate.Value

lcDevmode = IIF(EMPTY(Thisform.cDevmode),'',Thisform.cDevmode)

loXFF = Thisform.oPreviewCnt.oXFRXWriter

lcPrinter = STRTRAN(ALLTRIM(thisform.cboName.Value)," \ \","\\")

IF lnPrintRange = 1

   * We're printing all pages

   IF lnCollate = 1


FOR i = 1 TO lnCopies


   loXFF.PrintDocument(lcPrinter, "ExternalRpts", ;

                     '1-'+TRANSFORM(loXFF.PageCount), , lcDevmode)


ENDFOR

   ELSE


FOR lnCurPage = 1 TO loXFF.PageCount


   FOR i = 1 TO lnCopies


      loXFF.PrintDocument(lcPrinter, "ExternalRpts", ; 

                               TRANSFORM(lnCurPage), , lcDevmode)


   ENDFOR


ENDFOR 

   ENDIF 

ELSE

   * Now we're ready to print the selected pages

   IF lnCollate = 1


FOR i = 1 TO lnCopies


   loXFF.PrintDocument(lcPrinter, "ExternalRpts", ; 

                    TRANSFORM(lnPagesFrom)+'-'+TRANSFORM(lnPagesTo), , ;

                    lcDevmode)


ENDFOR 

   ELSE


FOR lnCurPage = lnPagesFrom TO lnPagesTo


   FOR i = 1 TO lnCopies



loXFF.PrintDocument(lcPrinter, "ExternalRpts", ; 

                        TRANSFORM(lnCurPage), , lcDevmode)


   ENDFOR 


ENDFOR 

   ENDIF 

ENDIF 

WAIT CLEAR 

Troubleshooting

If you have any experience with programming, you already know that things often don’t go as planned.  Here is something related to using XFRX that you may run into:
· When you don’t see something you’re expecting, it’s likely that an error was encountered and trapped by the preview container’s error method.  The XFRX preview container sends messages to the debugout window, and if you don’t have the debugout window visible, you won’t see the messages.  You may want to override the Error method of the container with your own error trapping.
Conclusion

XFRX has many additional features I haven’t been able to cover here.  For example, you can save reports directly to an XFF file and can open that file later for printing, previewing or even for further manipulation.  You can even convert an existing XFF file to another file type.
The DRAW method is very powerful, allowing you to manipulate reports both during and after the rendering process.  You can add tooltips, hyperlinks, bookmarks and watermarks to your reports.
When sending output to an Excel spreadsheet, the SETOTHERPARAMS method allows you much finer control over the formatting of the output.

I encourage you to review the XFRX documentation and play around with the evaluation version to discover the full capabilities XFRX has to offer.

Appendix A – Controlling a printer programmatically

Using the code from News2News as a starting point, I’ve come up with a VFP program that allows you to set specific properties of a printer without hard coding each setting and without requiring that you know how to use API programming. (The result doesn’t really resemble the code from News2News at all.)
The program is called SetPrinterProperties.prg and is included with the source code for this session.  The class it defines is called SetPrintProperties.  Instructions are included at the top of the program.  Use of this program does require you to be familiar with the Windows.h file, which is also included with the source code for this session.  Documentation about the various settings in the Windows.h file that pertain to printing can be found at:

http://msdn.microsoft.com/library/default.asp?url=/library/en-us/gdi/prntspol_8nle.asp 

A form called PrintTest.scx is also included in the source code for this session, and it has several examples of how to call the SetPrintProperties class.
The SetPrintProperties class currently only allows you to change printer settings and only supports properties in the Windows.h file from DM_ORIENTATION through DM_COLLATE.  The class will send your settings to the printer, but will also return the DEVMODE to you as a string in case you need to use it later, for example to send to the XFRX PrintDocument method. In order to use this program, you need to SET PROCEDURE TO ShowPrinterProperties, then instantiate it:
SET PRINTER TO NAME (GETPRINTER())

loSetPrinterProperties = CREATEOBJECT('SetPrintProperties')

Once you have a reference to the object, call the SetProperty method with the constants for the property and the value:

loSetPrinterProperties.SetProperty(DM_ORIENTATION, DMORIENT_LANDSCAPE)

Check that there weren’t any errors before sending reports to the printer:

IF loSetPrinterProperties.lError

   MESSAGEBOX(loSetProperty.cError)

ELSE 

   REPORT FORM Report1 TO PRINTER

   REPORT FORM Report2 TO PRINTER 

ENDIF 

Clear the object reference:

loSetPrinterProperties = NULL

The above example does not save the DEVMODE returned by the SetProperty method.  It is simply sending reports to the printer after the properties were changed. This example also changed properties for the current printer which was selected by the GETPRINTER() function.  The SetProperty method has an optional third parameter that takes the name of a printer, if you don’t want to change properties for the current default.  The example below shows how to use this third parameter:
loSetPrinterProperties = CREATEOBJECT('SetPrintProperties')

lcDevmode = loSetPrinterProperties.SetProperty(DM_DEFAULTSOURCE, ;

DMBIN_MANUAL, 'MyPrinter')

llSuccess = NOT loSetPrinterProperties.lError

lcError = loSetPrinterProperties.cError

loSetPrinterProperties = NULL

IF llSuccess

   * Do something here with lcDevmode

ELSE 

   MESSAGEBOX(lcError)

ENDIF 

Note that we could have used GETPRINTER() as the third parameter in the above example.
The SetPrintProperties class has a ResetPrinter method which allows you to change printer properties, print some reports, reset the printer, change some other properties, and do some more printing, all without release the reference to the class.  Here’s an example of how you would do this:

Let’s look at what’s happening in this code.  When you instantiate it, it saves the current printer and the current settings for the printer so we can reset them later.

FUNCTION Init()

LOCAL lcRptFile, lnOldSelect 

lnOldSelect = SELECT()

* Use a unique file name so we can use this in a multi-user situation

* Using a cursor instead of a physical file doesn't work, but we can

* create the FRX from a cursor.

lcRptFile = SYS(3)+".FRX"

CREATE CURSOR TempCur (Temp C (10))

CREATE REPORT (JUSTSTEM(lcRptFile)) FROM TempCur

USE IN TempCur

USE (lcRptFile) EXCLUSIVE ALIAS RptFile

SYS(1037,2)

This.cOrigExpr     = EXPR

This.cOrigDevnames = TAG

This.cOrigDevmode  = TAG2

This.cNewDevmode   = TAG2

This.cOrigPrinter  = SET("Printer",3)

* Get rid of the temporary FRX

USE IN RptFile

ERASE (JUSTSTEM(lcRptFile)+".*")

SELECT (lnOldSelect)

ENDFUNC 

The SetProperty method verifies the parameters, sets the current printer as the one you want to change, and calls the DoIt method:

FUNCTION SetProperty(tnPropertyPos, tnSetting, tcPrinter)

* Sets the property positioned in the tcPropertyPos of the DEVMODE

* structure to the setting in tcSetting.

* Will act on the printer named in tcPrinter, or the current printer if

* tcPrinter is empty.

* These parameters must values representing constants from the Windows.H file

* Will return the DEVMODE structure or an empty string if there's an error.

LOCAL lcPrinter, lcTemp

* Validate parameters

IF TYPE('tnPropertyPos') <> 'N' OR TYPE('tnSetting') <> 'N'

   This.lError = .T.

   This.cError = "You must send values for the first two parameters"

   RETURN ''

ENDIF 

IF EMPTY(tcPrinter)

   lcPrinter = SET("Printer",3)

ELSE

   lcPrinter = tcPrinter

   SET PRINTER TO NAME (lcPrinter)

ENDIF 

* Done with parameter validation.

This.DoIt(tnPropertyPos, tnSetting)

RETURN This.cNewDevmode

ENDFUNC 

The DoIt method is where the real work is done:

FUNCTION DoIt(tnPropertyPos, tnSetting)

LOCAL lcDevmode, lnFlags, lcRptFile, lnOldSelect, lnPropOffs, lnDC

lnFlags = CTOBIN(SUBSTR(This.cNewDevmode,41,4),'4sr')

IF BITAND(lnFlags,tnPropertyPOs) <> 0

   lnPropOffs = This.GetProp4DM(tnPropertyPos,'Offset')

   IF lnPropOffs <> 0


* Validate that the setting requested is supported for this printer.


* Note: In some cases this isn't necessary because the check against


* the flags field in DEVMODE indicated whether or not the property is


* available, e.g. Duplex, Collate.  The DeviceCapabilities would just


* return the same info.  Right now, the only check it does is for


* paper bins.


lnDC = This.GetProp4DM(tnPropertyPos,"DC")


DO CASE 


CASE lnDC < 0


   * Property Position not supported


   RETURN 


CASE lnDC = 0


   * Not sure how to validate


OTHERWISE


   * Confirm that the setting is available for the printer


   IF NOT This.CheckDC(lnDC, tnSetting)



RETURN 


   ENDIF 


ENDCASE 


This.cNewDevmode = STUFF(This.cNewDevmode, lnPropOffs, 2, ;

                                BINTOC(tnSetting,'2sr'))


* Write the changes to the printer


lnOldSelect = SELECT()


* Use a unique file name so we can use this in a multi-user situation


* Using a cursor instead of a physical file doesn't work, but we can


* create the FRX from a cursor.


lcRptFile = SYS(3)+".FRX"


CREATE CURSOR TempCur (Temp C (10))


CREATE REPORT (JUSTSTEM(lcRptFile)) FROM TempCur


USE IN TempCur


USE (lcRptFile) EXCLUSIVE ALIAS RptFile


replace expr WITH '', ;



  tag  WITH '', ;



  tag2 WITH This.cNewDevmode


SYS(1037,3)

&& Writes the printer settings out to the printer


* Get rid of the temporary FRX


USE IN RptFile


ERASE (JUSTSTEM(lcRptFile)+".*")


SELECT (lnOldSelect)

   ELSE


This.lError = .T.


This.cError = "Invalid property value sent as first parameter"


This.cNewDevmode = ''

   ENDIF 

ELSE

   This.lError = .T.

   This.cError = "That property not available for the printer"

   This.cNewDevmode = ''

ENDIF 

ENDFUNC 

Reviewing this in more detail, the first thing we do is call the CTOBIN function to convert the portion of the DEVMODE structure with the flags from a binary string to a decimal value.  
Knowing the decimal value for the bit that represents the property we want to change, we can use BITAND to make sure this property is available for the printer.
Once we confirm that the property is available, we call a function to determine what the offset is for that property in the DEVMODE structure.  Normally, I would use a formula to calculate the offset by determining what power of 2 the tnPropertyPos represents.  But because there are holes in the sequence of the values for the properties, it’s not quite as simple.  I decided to use a CASE statement that includes a CASE for every property and returns the offset.  In doing this, I can also use the same function to return the DC_ constant value for checking the DeviceCapabilities.
FUNCTION GetProp4DM(tnPropertyPos, tcProperty)

* Returns either the offset or the corresponding DeviceCapability value

* tnPropertyPos is the value of the bit for the property in the dmFields

* tcProperty is either "Offset" or "DC"

LOCAL lnOffset, lnDCValue

* The offset is the offset in the DEVMODE structure for the property

* The DCValue is the contant for the device capabilities from Windows.h

DO CASE 

CASE tnPropertyPos = 1


&& DM_ORIENTATION

   lnOffset = 45

   lnDCValue = 0

CASE tnPropertyPos = 2


&& DM_PAPERSIZE

   lnOffset = 47

   lnDCValue = 0

CASE tnPropertyPos = 4


&& DM_PAPERLENGTH

   lnOffset = 49

   lnDCValue = 0

CASE tnPropertyPos = 8


&& DM_PAPERWIDTH

   lnOffset = 51

   lnDCValue = 0

CASE tnPropertyPos = 16


&& DM_SCALE

   lnOffset = 53

   lnDCValue = 0

CASE tnPropertyPos = 256

&& DM_COPIES

   lnOffset = 55

   lnDCValue = 0

CASE tnPropertyPos = 512

&& DM_DEFAULTSOURCE

   lnOffset = 57

   lnDCValue = 6

CASE tnPropertyPos = 1024

&& DM_PRINTQUALITY

   lnOffset = 59

   lnDCValue = 0

CASE tnPropertyPos = 2048

&& DM_COLOR

   lnOffset = 61

   lnDCValue = 0

CASE tnPropertyPos = 4096

&& DM_DUPLEX

   lnOffset = 63

   lnDCValue = 0

CASE tnPropertyPos = 8192

&& DM_YRESOLUTION

   lnOffset = 65

   lnDCValue = 0

CASE tnPropertyPos = 16384

&& DM_TTOPTION

   lnOffset = 67

   lnDCValue = 0

CASE tnPropertyPos = 32768

&& DM_COLLATE

   lnOffset = 69

   lnDCValue = 0

OTHERWISE

   This.lError = .T.

   This.cError = "This program doesn't support the property " + ;

                  "position requested."

   lnOffset = 0

   lnDCValue = -1

ENDCASE 

RETURN IIF(tcProperty="DC",lnDCValue,lnOffset)

ENDFUNC 

After getting the offset in the DEVMODE structure, we should check that the setting is supported for the printer.  Currently, I’m only doing this for the paper bins.  You can see the code that does this in the CheckDC method.

FUNCTION CheckDC(tnValue, tnSetting)

* This function is currently only working for checking paper bins

LOCAL lnIndex, lcPrinterName, lnBuffer, lcPort, lcIdBfr, lnDCCnt, lnDCID

DECLARE INTEGER DeviceCapabilities IN winspool.drv;

   STRING pDevice, STRING pPort, INTEGER fwCapability,;

   STRING @pOutput, INTEGER pDevMode

lcPrinterName = SET("Printer",3)

APRINTERS(laAvailPrinters)

FOR i = 1 TO ALEN(laAvailPrinters,1)

   IF UPPER(laAvailPrinters[i,1]) = UPPER(lcPrinterName)

lcPort = laAvailPrinters[i,2]


EXIT 

   ENDIF 

ENDFOR

* First call DeviceCapabilities in a way that returns the size of

* the buffer we need

* lnBuffer = DeviceCapabilities(lcPrinterName, lcPort, tnValue, 0, 0)

* The above call is only returning 2.  Need to figure out why

* it's not correct.

lnBuffer = 1024

lcIDBuf = REPLICATE(CHR(0), lnBuffer)

lnDCCnt = DeviceCapabilities(lcPrinterName, lcPort, tnValue, @lcIdBuf, 0)

FOR i = 1 TO lnDCCnt

   IF CTOBIN(SUBSTR(lcIdBuf, (i*2)-1, 2),'2sr') = tnSetting


RETURN .T.

   ENDIF 

ENDFOR

This.lError = .T.

This.cError = "That setting not supported"

RETURN .F.

ENDFUNC 

Now we need to write the new setting into the correct position of the DEVMODE structure.  The setting was sent as a decimal number, so we need to convert that into a binary string and stuff it into the DEVMODE binary string.

At this point, we just need to use the same technique with SYS(1037) that I showed previously and the DEVMODE structure is sent to the printer.
When you destroy the object, the ResetPrinter method will be called automatically to return the printer settings as they were.  This will not only set the properties back as they were, but will SET PRINTER TO the default printer from before the class was instantiated.  Note that even if you didn’t change the default printer, but sent a different printer as the third parameter to SetProperty, the program had to change the default printer in order to send the changes to it.  It’s not possible for the program to SET PRINTER back to the original printer as part of the SetProperty method, or the changes written will be lost.

Lisa Slater Nicholls has advised me that when you have a situation like this where the user can make a series of changes, those changes are saved in a stack.  In order to restore the printer properly, you must undo each set of changes in the reverse order that they were made.  However, in my testing I’ve found that by saving the original state of the printer and restoring from that set of settings, all changes made are reverted with one step.  Still, I haven’t tested every possibility, so be aware of this in case you run into something contrary to my findings.
Verifying the setting

http://msdn.microsoft.com/library/default.asp?url=/library/en-us/gdi/prntspol_21bn.asp
Appendix B – A primer on bit manipulation

Although I’m sure most of you are at least familiar with binary numbers, some of you may need a refresher course on bit manipulation.  If you are quite familiar with the use of BITOR, BITAND and BITTEST functions in VFP, you don’t need to read this section.
Byte basics

First, a byte is made up of eight bits, and these bits are numbered from right to left.  The textbox below shows a single byte with all of the bits turned on, i.e. they are set to 1.  Bits that are off are set to 0.


   |        |       |       |       |        |       |        |                  

128    64    32   16     8       4      2      1

The example above is the binary representation of the number decimal number 255.  We arrived at this by adding the representation of each bit:


1+2+4+8+16+32+64+128 = 255
If any of these bits were 0, we wouldn’t include that number in the sum.  For example…

…would be 223 because the bit for 32 is 0 and so the math would look like:


1+2+4+8+16+64+128 = 223

Of course, when you view a binary number from within VFP, you don’t see a series of 1s and 0s.  You see funny characters instead.  VFP can’t display binary, so if you need to view the number, you must convert it to decimal first.  The CTOBIN() function in VFP will do this for you.

Words and double words

Binary words are two bytes long, and double words are four bytes.  These would follow the same rules as above, with the additional bytes being added to the left.
Bit comparisons

VFP also has some built-in functions that allow us to easily test or compare bits.

BITTEST

The BITTEST function does exactly what you would expect.  You pass it the decimal value of byte, word or double word, plus the position from the right that represents the bit you want to test, and BITTEST will return .T. or .F. based on whether that bit is 1 or 0.  Using the numbers 255 and 223, as represented by the two examples of bytes shown above, let’s look at the fifth bit of each byte.  (Numbering starts with 0, so the first bit is 0, the second is 1, etc.)

BITTEST(255,5) returns .T.

BITTEST(223,5) returns .F.

If you know the position of the bit that you want to test, this is a very useful function.

BITAND and BITOR

Both BITAND and BITOR do a comparison between two bytes, words or double words and produce a third byte, word or double word representing a bit-by-bit comparison of the two sources.  For example, let’s use these two bytes as an example:


The first byte is 2+8+32+128 = 170

The second is 16+32+64+128 = 240

BITAND will compare each bit in the two numbers passed.  If both bits in any give position are 1, that position will be 1 in the result.  If either of them are 0, the result is 0.  So, BITAND(170,240) results in the following binary representation:


The decimal value is 32+128 = 160

The BITOR also compares each bit in the two numbers passed, but if either of the bits for any given position is 1, that position will be 1 in the result.  The position will only be 0 if BOTH bits are 0.  So BITOR(170,240) results in the following binary representation:


The decimal value is 2+8+16+32+64+128 = 250
Putting it into practice

Using this knowledge of how BITAND and BITOR work, you can see why I determined that using BITOR would not determine whether or not a bit was set.  For the second parameter, we pass a decimal value that represents the property we are interested in.  That decimal value will translate into a binary value that has only a single bit turned on.  Using this value with a BITOR will always result in a binary value that has that bit is turned on.  Using BITAND, however, will either return 0 or the value of the property we are interested in.  If the result is 0, that bit is not on in the first parameter and therefore the property is not available for the printer.  If the result matches the second parameter (i.e. the value for the property we want to change), then that property is turned on in the flags field and is available for that printer.
Using BINTOC and CTOBIN
BINTOC and CTOBIN have been part of VFP since version 5, but with VFP 9 some enhancements were added.  Originally, the only possible values for the second parameter were the size in bytes of the binary number.  BINTOC and CTOBIN were originally intended to be used in index expressions, which require the least significant bit be on the left and that the sign bit be toggled.  VFP 9 was enhanced so that BINTOC and CTOBIN could be used to convert or create standard binary strings.  The second parameter was changed to accept character data as well as a numeric value.  This means you can specify the size of the binary string within the same string as other non-numeric flags.  In order to use BINTOC and CTOBIN for non-index strings, you must use the “R” and “S” flags to reverse the order used by index tags and to prevent the sign bit from being toggled.  This is why you see “2sr” or “4sr” used with these functions in the SetPrinterProperties code.

� If you are unfamiliar with bit manipulation, please review “Appendix B – A primer on bit manipulation” at the end of this document.





